

全国测绘地理信息职业教育教学指导委员会 推荐



工程测量计算与平差

李泽球 主编

基于Excel

基于Excel

基于Excel

基于Excel

基于Excel

基于Excel



WUTP

武汉理工大学出版社

Wuhan University of Technology Press

工程测量计算与平差

主 编 李泽球



武汉理工大学出版社

· 武 汉 ·

内 容 简 介

本书共 13 章,首先介绍了在 Excel 中进行路线高程计算、交会测量计算、导线测量计算、高斯投影及投影换带计算、坐标变换的基本公式和方法;详细介绍了在 Excel 中进行高程控制网、三角网、测边网、导线网和边角网平差计算的基本公式和方法;最后介绍了误差椭圆的计算与绘制方法。本书充分利用 Excel 的函数功能,针对工程测量的各种基本计算和各类工程控制网的平差计算,设计了大量便捷有效的组合计算公式,使工程测量与平差计算变得简单易行。其中,边角网、导线网方差分量估计的方法为边角控制网数据处理提供了一条便捷的途径,也充分体现了 Excel 用于工程测量和平差计算的优越性。本书基本涵盖了工程测量常规的计算内容,示例丰富、过程详细、可操作性强,并附有大量练习题。本书可以作为高职高专测绘工程专业的教材,也可以作为工程测量技术人员的参考用书。

图书在版编目(CIP)数据

工程测量计算与平差/李泽球主编. —武汉:武汉理工大学出版社,2015.5
ISBN 978-7-5629-4351-8

I. ①工… II. ①李… III. ①工程测量-测量平差-高等职业教育-教材
IV. ①TB22

中国版本图书馆 CIP 数据核字(2015)第 119389 号

项目负责人:汪浪涛

责任编辑:张莉娟

责任校对:余土龙

装帧设计:许伶俐

出版发行:武汉理工大学出版社

地 址:武汉市洪山区珞狮路 122 号

邮 编:430070

网 址:<http://www.techbook.com.cn>

经 销:各地新华书店

印 刷:湖北恒泰印务有限公司

开 本:787×960 1/16

印 张:12.75

字 数:235 千字

版 次:2015 年 6 月第 1 版

印 次:2015 年 6 月第 1 次印刷

定 价:25.00 元

凡购本书,如有缺页、倒页、脱页等印装质量问题,请向出版社发行部调换。

本社购书热线:027-87515778 87515848 87785758 87165708(传真)

· 版权所有,盗版必究 ·

前 言

近几年来,随着国家和地方大规模工程建设的开展,高职高专的测绘工程专业得到迅猛发展。目前,全国有近百所高职高专学校开办了测绘工程专业或类似专业,该专业的基本技能是测、算、绘。随着教育的不断进行,各校都加强了实践性教学环节,学生的专业动手能力明显得到提高。但目前用人单位普遍反映毕业生测、绘尚可,计算偏弱。由于测绘工程专业就是要与数据打交道,因此测量数据的计算处理能力十分重要。本书的编写目的就是为测绘工程专业学生提供一本提高计算能力的参考书。此外,应用 Excel 进行工程测量计算和平差时,既不需要专业软件,也不使用算法语言,只需应用 Excel 函数,便捷灵活、效率高。例如,在测边网平差计算中,一个简单公式的拖动,就可以完成 N 个边长观测误差方程式系数的计算;导线网、边角网的方差分量估计问题,在 Excel 中可以轻松解决。笔者在长期的专业教学实践中,深感 Excel 用于工程测量和平差计算有显著的优越性,借此与广大同行分享教学实践经验。

本书主要介绍与工程测量相关的计算理论和方法,以及在 Excel 中实施操作的过程,其中包括笔者精心设计的大量组合函数的应用,示例丰富完整、可操作性强。

需要说明的是:本书中有关计算表格的安排、组合函数的设计、平差计算检核的方法及各部分具体计算思路纯属个人意见,并非十分完善,在此仅抛砖引玉,以期促进 Excel 在工程测量和平差计算中更广泛的应用。书中若有不当之处,诚请读者批评指正。

编 者
2014.5

目 录

1	Excel 相关函数	1
1.1	数学和三角函数	2
1.2	逻辑函数	7
1.3	文本函数	8
1.4	查找与引用函数	9
1.5	统计函数.....	12
2	路线高程计算	17
2.1	水准路线的高程计算.....	17
2.2	三角高程路线计算.....	19
3	交会测量计算	24
3.1	前方交会计算.....	24
3.2	后方交会计算.....	26
3.3	测边交会计算.....	30
4	导线测量计算	34
4.1	附和导线计算.....	34
4.2	闭合导线计算.....	38
4.3	无定向导线计算.....	39
4.4	单结点导线计算.....	41
5	高斯投影及换带计算	48
5.1	高斯投影坐标正算.....	48
5.2	高斯投影坐标反算.....	49
5.3	子午线收敛角计算.....	50
5.4	换带计算.....	53
6	坐标变换	56
6.1	平面坐标变换的基本公式.....	56
6.2	施工坐标变换为测图坐标.....	57
6.3	测图坐标变换为施工坐标.....	58
6.4	断面数据转换.....	59
7	测量平差原理	64

7.1	间接平差的数学模型	65
7.2	间接平差的矩阵表达	66
7.3	精度评定	67
8	高程网平差	73
8.1	水准网测量间接平差原理	73
8.2	水准网平差计算步骤	74
8.3	水准网平差示例	75
8.4	水准网平差计算过程的优化	81
8.5	三角高程网间接平差	89
9	三角网平差	96
9.1	方向误差方程式	96
9.2	近似坐标计算	98
9.3	三角网的闭合差计算及检查	99
9.4	三角网方向的平差步骤	100
10	测边网平差	123
10.1	边长误差方程式的列立	123
10.2	测边网近似坐标计算	124
10.3	测边定权	125
10.4	测边网闭合差的类型	125
10.5	测边网间接平差示例	126
11	导线网平差	140
11.1	导线网方向误差方程式	140
11.2	导线网边长误差方程式	140
11.3	导线网近似坐标的计算	141
11.4	导线网定权	141
11.5	边长误差方程式单位权化	142
11.6	导线网闭合差的类型	142
11.7	导线网方差分量估计	143
12	边角网平差	162
13	误差椭圆	174
13.1	误差椭圆参数	174
13.2	相对误差椭圆	175
13.3	误差椭圆绘制	176
	参考答案	183
	参考文献	197

1 Excel 相关函数

Excel 是一种应用广泛的办公软件,具有丰富的数据处理功能。在 Excel 中进行工程测量和平差计算具有以下优点:

(1) Excel 是 Windows 操作系统中的常用应用软件,应用 Excel 来进行测量数据处理十分方便。

(2) Excel 具有丰富的函数功能,应用这些函数进行测量数据处理,能大大提高工作效率。

(3) Excel 能进行矩阵运算,这给控制网平差计算带来了极大的方便。例如,在矩阵运算中,法方程式的组成和解算只需要输入两个极其简单的矩阵公式就能立刻完成。

(4) Excel 的拖动计算功能、复制粘贴功能对于大量重复计算十分有用。例如组成误差方程式时,只需组成第一个观测误差方程式,对其他 $n-1$ 个误差方程式只需拖动即可完成(要设计一个适当的输入公式才能做到这一点)。在控制网平差计算中,近似坐标计算、三角网闭合差计算、平差坐标计算等都可以充分利用这个功能。

(5) 在 Excel 中采用表格形式进行工程测量和平差计算,在形式上与传统的手算表格形式非常接近,清晰直观,便于检查和修改。平差计算过程完全按传统步骤,以常用平差公式为基础,非常适合初学者学习、掌握和理解平差原理。

(6) 在 Excel 中能很方便地进行方差分量估计。方差分量估计的目的在于确定控制网的精度指标不依赖人为的因素,不依赖先验方差分量估计的结果,仅取决于控制网观测的精度,并且只有唯一的结果。一般情况下,方差分量估计只有采用专用软件或自编应用程序才能进行。在 Excel 中,只需要输入相应的计算公式,利用 Excel 的自动重算功能即可快速完成方差分量估计。

(7) 在 Excel 中进行控制网平差计算具有很强的灵活性,适用于工程测量范围内的各级各类控制网的严密平差,其中包括三角网、导线网、测边网、边角网、线形锁、混合网等。

在 Excel 中进行工程测量和平差计算需要用到 Excel 函数。Excel 函数共包含 10 类,分别是常用函数、财务函数、日期与时间函数、数学和三角函数、统计函数、查询和引用函数、数据库函数、文本函数、逻辑函数以及信息函数。本章主要介

绍与工程测量和平差计算有关的函数。掌握这些函数的应用,将使工程测量和平差计算变得简单而有趣。

1.1 数学和三角函数

1.1.1 PI() 函数(π 函数)

PI()

该函数是一个常数函数,无自变量,返回 π 值。

例如,在单元格输入: =PI() √(确认),该单元格显示:3.141593(显示位数与单元格的宽度和单元格格式有关)。

1.1.2 INT 函数(下取整函数)

INT(number)

该函数的返回数值向下取整为最接近的整数,函数中的 number 为被取整的数值。对于正数,向下取整就是去掉小数部分;对于负数,向下取整就是去掉小数部分,并使整数部分的绝对值增大 1。

例如,在单元格中输入: =INT(5.9) √(确认),单元格显示:5;在单元格中输入: =INT(-5.9) √(确认),单元格显示:-6。

1.1.3 MOD 函数(取余函数)

MOD(number, divisor)

该函数返回两数相除的余数,函数中 number 为被除数,divisor 为除数。当除数设定为 1,被除数为正数时,返回小数部分。

例如,在单元格中输入: =MOD(PI(),1) √(确认),单元格显示:0.141593。

该函数中被除数为负时,返回的值为:负余数+除数。在测量坐标反算中,反算的方位角有时是负数。应用这个函数就能很方便地将计算结果化成正方位角。

例如,将-2.2469(弧度)的方位角化成正方位角,在单元格中输入: =MOD(-2.2469,2*PI()) √(确认),单元格显示:4.036285,这就是以正值表示的该方位角,单位仍然是弧度。

1.1.4 DEGREES 函数(弧化度函数)

DEGREES(angle)

该函数返回以度($^{\circ}$)为单位的角度值,函数中的 angle 为以弧度为单位的角度

值。在 Excel 中进行运算时,角度的单位是弧度。对运算角度的结果有时需要转化成以度为单位的角度,或者以度分秒表示的角度,以便检查核对。

例如,将以弧度为单位的角度值 0.870356 转化成以度为单位的角度值。在单元格输入: =DEGREES(0.870356) √(确认),则该单元格显示:49.86773。即 $0.870356(\text{弧度})=49^{\circ}86773$ 。

若要显示以度分秒表示的角度,则要设计一个稍为复杂的表达式。

例如,将以弧度为单位的角度值 0.870356 转化为以度分秒表示的角度值。在单元格 A1 中输入:0.870356,在单元格 B1 中输入: =INT(DEGREES(A1))+INT(MOD(DEGREES(A1),1)*60)/100+MOD(DEGREES(A1)*60,1)*3/500 √(确认),则单元格 B1 中显示:49.520381。

这个数字格式是自行定义的:小数点前为度数,小数点后两位为分数,分数后两位为秒数的整数部分,再以后为秒数的小数部分。这个显示结果就是 $49^{\circ}52'03''81$ 。

1.1.5 RADIANS 函数(度化弧函数)

RADIANS(angle)

该函数返回以弧度为单位的角度值,函数中 angle 必须是以度为单位的角度值。

例如,将以度为单位的角度值 49.86773 转化以弧度为单位的角度值。在单元格中输入: =RADIANS(49.86773) √(确认),则该单元格显示:0.870356。即 $49^{\circ}86773=0.870356(\text{弧度})$ 。

测量中角度的观测值都是以度分秒表示的,若这些观测值要在 Excel 中参与运算,必须先将它们转化成弧度值。进行这种转化也要设计一个稍为复杂的表达式。

例如,将以度分秒表示的角度值 $49^{\circ}52'03''81$ 转化成以弧度为单位的角度值。在单元格 C1 中输入:49.520381(自定义格式,小数点前为度,小数点后两位为分数,再后两位为秒数,小数点后第五位起为秒的小数部分),在单元格 D1 中输入: =RADIANS(INT(C1)+INT(MOD(C1,1)*100)/60+MOD(C1*100,1)/36) √(确认),则单元格 D1 中显示:0.870356。

1.1.6 SUM 函数(和函数)

SUM(number1,number2,...)

该函数返回单元格区域中所有数值之和,number1,number2,...可以是参与求和的不同数值,也可以是单元格区域。

例如,在单元格 E1 中输入: =SUM(2,5,8,17) √(确认),单元格 E1 中显示:32。若输入: =SUM(A2:A6) √(确认),则单元格 E1 显示由 A2 到 A6 共 5 个单元格的数值之和。

1.1.7 SUMSQ 函数(平方和函数)

SUMSQ(number1,number2,...)

该函数返回所有参数的平方和,参数可以是数值、数组、名称,或者是对数值单元格的引用。

例如,在单元格 F1 中输入: =SUMSQ(2,5,8,11) √(确认),则单元格 F1 显示:214。若输入: =SUMSQ(B2:B5) √(确认),则单元格 F1 显示由 B2 到 B5 共 4 个单元格的数值平方之和。

1.1.8 SQRT 函数(开方函数)

SQRT(number)

该函数返回数值的平方根。

例如,在单元格 G1 中输入: =SQRT(8) √(确认),则单元格 G1 显示: 2.828427。

1.1.9 ATAN2 函数(方位角反算函数)

ATAN2(x,y)

该函数根据给定的 X 轴和 Y 轴坐标值(x,y)返回反正切值。这里的 x,y 是指点在坐标系中的坐标,函数值是点到坐标原点的边与 X 正轴的夹角。若求坐标系中两点间的方向,则函数中的坐标值(x,y)用坐标增量代替。在测量计算中,由两个已知点的坐标反算坐标方位角,用这个函数计算很方便。

【例 1.1】 已知 A、B 两点的坐标增量: $\Delta X_{AB}=236.457\text{ m}$, $\Delta Y_{AB}=-109.286\text{ m}$, 求 AB 的坐标方位角 α_{AB} 。

【解】 如图 1.1 所示,在单元格 A1 中输入 236.457,在单元格 A2 中输入 -109.286,在单元格 A3 中输入: =ATAN2(A1,A2) √(确认),则单元格 A3 显示: -0.432938。这是以弧度表示的 AB 的方位角,但是个负数。测量中方位角一般以正值表示,故需要用取余函数换成正值,如图 1.1 所示。

在单元格 A4 中输入: =MOD(A3,2 * PI()) √(确认),单元格 A4 显示: 5.8502477。

上面两步计算的过程可以合并成一步完成。在单元格 A5 中输入: =MOD

	A	B	C	D	E	F	G	H	I	J
1	236.457									
2	-109.286									
3	-0.432938	ATAN2(A1, A2)								
4	5.8502477	MOD(A3, 2*PI())								
5	5.8502477	MOD(ATAN2(A1, A2), 2*PI())								
6	335.11402	INT(DEGREES(A5))+INT(MOD(DEGREES(A5), 1)*60)/100+MOD(DEGREES(A5)*60, 1)*3/500								
7	5.8502477	RADIANS(INT(A6)+INT(MOD(A6, 1)*100)/60+MOD(A6*100, 1)/36)								
8										
9										
10										

图 1.1 ATAN2 函数的应用

(ATAN2(A1, A2), 2 * PI()) √(确认), 单元格 A5 显示: 5.8502477。

以弧度表示的方位角不便于检查核对, 再将其转换成度分秒表示的方位角。应用前述的转换表达式, 在单元格 A6 中输入: =INT(DEGREES(A5))+INT(MOD(DEGREES(A5), 1)*60)/100+MOD(DEGREES(A5)*60, 1)*3/500 √(确认), 单元格 A6 显示: 335.11402, 即 AB 的方位角为 335°11'40"2。

为了消除错误, 检验计算的可靠性, 用度分秒的形式转化弧度的表达式, 再将这个角度转化为弧度表示的角度。在单元格 A7 中输入: =RADIANS(INT(A6)+INT(MOD(A6, 1)*100)/60+MOD(A6*100, 1)/36) √(确认), 单元格 A7 显示: 5.8502477。证明计算无误。

在计算过程中发现, 这个公式对于整十倍秒数的角度(角度的秒数为 00"、10"、20"、...、50", 无小数)转换有时会出错。为了避免这种情况出现, 在输入角度或方向值时, 对整十倍秒数的角度或方向加 0"001 输入。

以上这两个转换式比较复杂, 输入麻烦、费时。但不必经常输入, 第一次使用时输入一次, 以后需要时采用复制、粘贴的方式比较方便。为了便于比较和认识, 在图 1.1 的单元格 B3 到 B7 中, 分别输入了对应于单元格 A3 到 A7 不带等号的函数表达式。

1.1.10 TRANSPOSE 函数(转置函数)

TRANSPOSE(array)

该函数返回转置单元格区域, 函数中 array 为所选择的需要转置的单元格区域。这是一个数组函数, 在输入函数前, 要选中结果的显示区域(多个单元格), 而不是一个单元格。函数输入完后, 需要同时按 Shift、Ctrl、Enter 三键确认, 而不是单击 √(确认)。这个函数常用于求转置矩阵。

【例 1.2】 某水准网按间接平差列出的误差方程式 $V = BX + L$ (见例 7.1) 如下所示:

$$\begin{pmatrix} v_1 \\ v_2 \\ v_3 \\ v_4 \\ v_5 \end{pmatrix} = \begin{pmatrix} 1 & 0 & 0 \\ 1 & -1 & 0 \\ 0 & -1 & 0 \\ 0 & 1 & -1 \\ 0 & 0 & -1 \end{pmatrix} \begin{pmatrix} x_1 \\ x_2 \\ x_3 \end{pmatrix} + \begin{pmatrix} -78.541 \\ -1.579 \\ 76.984 \\ -0.664 \\ -76.330 \end{pmatrix}$$

应用矩阵转置函数计算误差方程式系数矩阵 B 的转置矩阵 B^T 。

【解】 先将 B 矩阵中的 5 行 3 列元素分别输入到 A2 : C6 的单元格区域, 如图 1.2 所示。欲求矩阵的转置矩阵 B^T , 应先选中 B^T 的显示单元格区域。由于矩阵 B 的行列为 5×3 , 转置后的行列应为 3×5 。选中单元格区域 D2 : H4, 输入: =TRANSPOSE(A2 : C6), 同时按 Shift、Ctrl、Enter 三键确认, 则所选区域显示转置矩阵结果如图 1.2 所示。

D2		={=TRANSPOSE(A2:C6)}										
	A	B	C	D	E	F	G	H	I	J	K	
1	B			B ^T								
2	1	0	0	1	1	0	0	0				
3	1	-1	0	0	-1	-1	1	0				
4	0	-1	0	0	0	0	-1	1				
5	0	1	-1									
6	0	0	1									
7												

图 1.2 TRANSPOSE 函数应用

需要说明的是, 在 Excel 函数表中, TRANSPOSE 函数属于查找与引用函数, 因其常用于矩阵计算, 故在此介绍。

1.1.11 MMULT 函数(矩阵乘函数)

MMULT(array1, array2)

该函数返回两数组矩阵的乘积, 函数中 array1, array2 分别为两个数组矩阵, array1 的列数应与 array2 的行数相等。结果矩阵的行数与 array1 相等, 列数与 array2 相等。MMULT 函数也是数组函数。

【例 1.3】 根据例 1.2 的数据, 试计算法方程式的系数矩阵 $B^T P B$ 。

【解】 法方程式的系数矩阵 $B^T P B$, 是 3 个矩阵相乘, 且有 1 个要转置。可以一步一步地计算, 也可以应用嵌套函数一步算出。应用嵌套函数一步计算时, 必须先确定结果矩阵的行列数, 以便确定显示区域的大小。 B^T 的行列为 3×5 , P 的行列为 5×5 , B 的行列为 5×3 , 则 $B^T P B$ 的行列为 3×3 。利用例 1.2 的 B 矩阵和 B^T 矩阵, 在单元格区域 I2 : M6 中输入观测权矩阵(数据来源于例 7.1), 选中单元格区域 N2 : P4, 输入: =MMULT(D2 : H4, MMULT(I2 : M6, A2 : C6)), 同时按

Shift、Ctrl、Enter 三键确认,则所选区域显示法方程系数矩阵,结果如图 1.3 所示。

N2		={MMULT(D2:H4,MMULT(I2:M6,A2:C6))}																	
	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R	S
1		B			B ^T				P				B ^T PB						
2	1	0	0	1	1	0	0	0	2	0	0	0	0	3	-1	0			
3	1	-1	0	0	-1	-1	1	0	0	1	0	0	0	-1	4	-2			
4	0	-1	0	0	0	0	-1	1	0	0	1	0	0	0	-2	3			
5	0	1	-1						0	0	0	2	0						
6	0	0	1						0	0	0	0	1						

图 1.3 MMULT 函数的应用

应用嵌套函数计算本题时还可以省去计算转置一步。如直接在单元格区域 N2 : P4 中输入: =MMULT(TRANSPOSE(A2 : C6),MMULT(I2 : M6,A2 : C6)),同时按 Shift、Ctrl、Enter 三键确认,则所选区域显示结果与图 1.3 相同。

1.1.12 MINVERSE 函数(矩阵求逆函数)

MINVERSE(array)

该函数的返回数组代表逆矩阵。MINVERSE 函数也是一个数组函数。

【例 1.4】 计算例 1.3 求出的法方程系数矩阵的逆矩阵。

【解】 先在单元格区域 A1 : C3 中输入法方程系数矩阵,再选中单元格区域 D1 : F3,输入: =MINVERSE(A1 : C3),同时按 Shift、Ctrl、Enter 三键确认,则所选区域显示法方程系数矩阵的逆矩阵,如图 1.4 所示。

D1		={MINVERSE(A1:C3)}					
	A	B	C	D	E	F	G
1	3	-1	0	0.380952	0.142857	0.095238	
2	-1	4	-2	0.142857	0.428571	0.285714	
3	0	-2	3	0.095238	0.285714	0.52381	
4							

图 1.4 矩阵求逆

工程测量和平差计算中还要经常用到正弦函数、余弦函数、正切函数等,但因为比较简单,此处就不再介绍了。

1.2 逻辑函数

Excel 的逻辑函数在工程测量和平差计算中用得比较少,这里只介绍 IF 函数的应用。

IF 函数(判断选择函数)

IF(logical-test,value-if-true,value-if-false)

该函数判断一个条件是否满足,如果满足返回一个值;如果不满足则返回另一个值。在 IF 函数中,logical-test 为设定的条件,value-if-true 为满足条件返回的值,value-if-false 为不满足条件返回的值。

【例 1.5】 根据学生分数来评定学生成绩的等级。

【解】 如图 1.5 所示,在单元格区域 A2:A11 中分别输入 10 名学生的分数,在单元格 B2 中输入: =IF(A2<60,“不及格”,IF(A2<75,“及格”,IF(A2<85,“良好”,“优秀”))) √(确认),则单元格 B2 显示:良好。将单元格 B2 的公式一直拖动至 B11,则单元格区域 B2:B11 显示结果如图 1.5 所示。

	A	B	C	D	E	F	G	H
1	分数	等级						
2	75	良好						
3	56	不及格						
4	82	良好						
5	91	优秀						
6	78	良好						
7	64	及格						
8	94	优秀						
9	69	及格						
10	70	及格						
11	88	优秀						
12								

图 1.5 IF 函数的应用

1.3 文本函数

1.3.1 LEFT 函数(取左函数)

LEFT(text,num-chars)

该函数从一个文本字符串的第一个字符开始,返回指定个数的字符。函数中 text 为所选定的文本字符串,num-chars 为指定返回字符的个数,此项省略时,则返回所选文本的第一个字符。

例如,在单元格 A1 中,输入:ABCD,在单元格 B1 中,输入: =LEFT(A1,3) √(确认),则单元格 B1 显示:ABC。在单元格 C1 中,输入: =LEFT(A1) √(确认),则单元格 C1 显示:A。

1.3.2 RIGHT 函数(取右函数)

RIGHT(text,num-chars)

该函数从一个文本字符串的最后一个字符开始,返回指定个数的字符。函数

中 text 为所选定的文本字符串, num-chars 为指定返回字符的个数, 此项省略时, 则返回所选文本的最后一个字符。

例如, 在单元格 A2 中, 输入: ABCD, 在单元格 B2 中, 输入: =RIGHT(A2,2) √(确认), 则单元格 B2 显示: CD。在单元格 C2 中, 输入: =RIGHT(A2) √(确认), 则单元格 C2 显示: D。

在平差计算中, 组成误差方程式时会用到这两个文本函数。

1.4 查找与引用函数

1.4.1 ROW 函数(行号函数)

ROW(reference)

该函数返回一个引用的行号, 函数中 reference 为引用的单元格。

例如, 在单元格中输入: =ROW(C7) √(确认), 则该单元格显示: 7。

与 ROW 函数相近的还有一个 ROWS(array) 函数, 返回一个引用或数组的行数, 函数中 array 为引用或数组的单元格。例如, 在单元格中输入: =ROW(C3 : C7) √(确认), 则该单元格显示: 5。

1.4.2 COLUMN 函数(列号函数)

COLUMN(reference)

该函数返回一个引用的列号, 函数中 reference 为引用的单元格。

例如, 在单元格中输入: =COLUMN(C7) √(确认), 则该单元格显示: 3。

ROW 函数和 COLUMN 函数可在公式设计中用作步进变量。

与 COLUMN 函数相近的还有一个 COLUMNS(array) 函数, 返回一个引用或数组的列数, 函数中 array 为引用或数组的单元格。例如, 在单元格中输入: =COLUMN(C3 : F7) √(确认), 则该单元格显示: 4。

1.4.3 INDEX 函数(指定函数)

INDEX(array, row-num, column-num)

该函数在给定的单元格区域中, 返回特定行列交叉处单元格的值或引用。函数中 array 为给定的单元格区域, 也可以是数组常量, row-num 为选定的行号, column-num 为选定的列号。当 row-num 和 column-num 均不为零时, 函数返回一个特定单元格的数值。row-num 和 column-num 不能同时为零, 有一个为零时, 则返回给定区域的一个整行或一个整列。此时, INDEX 函数应按数组函数进行操作。

例如,在单元格中输入: =INDEX(A1:C3,2,3) √(确认),则该单元格显示 C2 单元格的数值。在单元格区域 D1:D3 中输入: =INDEX(A1:C3,0,2) 同时按 Shift、Ctrl、Enter 三键确认,则所选区域显示单元格 B1、B2、B3 的数值。

1.4.4 VLOOKUP 函数(列查找函数)

VLOOKUP(lookup-value, table-array, col-index-num, range-lookup)

使用 VLOOKUP 函数可以在表格或数值数组的首列查找指定的数值,并由此返回表格或数值数组当前行中指定列数处的单元格数值。函数中 lookup-value 为指定的数值,也可以是某个单元格的引用; table-array 为查找范围或区域; col-index-num 为 table-array 中的一个指定列数; range-lookup 为一个关于精密匹配(FALSE)或是近似匹配(TRUE 或省略)的逻辑值,在查找数值中,通常选择 FALSE。例如,公式 VLOOKUP(EF, A2:D11, 4, FALSE) 的作用就是在单元格区域 A2:D11 内查找首列为 EF 的行,返回该行第 4 列的数值或引用,精密匹配。

【例 1.6】 已知 A、B、C、D 四个点的平面坐标,试计算 AB、AC、AD、BC、BD、CD 的坐标方位角 α (以度分秒表示)。已知点的点号和坐标数据已输入表格,如图 1.6 所示。

E2		=VLOOKUP(RIGHT(\$D2), \$A\$2:\$C\$5, 2, FALSE) - VLOOKUP(LEFT(\$D2), \$A\$2:\$C\$5, 2, FALSE)								
	A	B	C	D	E	F	G	H	I	J
1	点号	X	Y	方向	ΔX	ΔY	α (弧度)	α (度分秒)	α (弧度)	
2	A	2321.652	7566.548	AB	116.927	104.774	0.7306359	41.51445	0.7306359	
3	B	2438.579	7671.322	AC	-313.908	182.456	2.61508122	149.49592	2.61508122	
4	C	2007.744	7749.004	AD	-217.993	121.709	2.63238715	150.49288	2.63238715	
5	D	2103.659	7688.257	BC	-430.835	77.682	2.96320364	169.46446	2.96320364	
6				BD	-334.92	16.935	3.09107137	177.06192	3.09107137	
7				CD	95.915	-60.747	5.71860971	327.39079	5.71860971	
8										

图 1.6 VLOOKUP 函数的应用

【解】 分别在 D1:I1 的单元格区域输入表头:方向、 ΔX 、 ΔY 、 α (弧度)、 α (度分秒)、 α (弧度);再在“方向”下各行分别输入要计算的方向编号。最后的 α (弧度)为检核计算。

(1) 计算 X 的坐标增量

在单元格 E2 中输入: =VLOOKUP(RIGHT(\$D2), \$A\$2:\$C\$5, 2, FALSE) - VLOOKUP(LEFT(\$D2), \$A\$2:\$C\$5, 2, FALSE) √(确认), 单元格 E2 显示:116.927。

这个公式的功能是以方向编号为参数,让 Excel 自动到指定的单元格区域去找这两个点的 X 坐标,并求其差。 $\$A\$2:\$C\5 就是指定的单元格区域;函数中的第三个参数为 2,是在指定的单元格区域内找第二列的元素,即 X 坐标;

RIGHT(\$D2)和 LEFT(\$D2)为查找值,即先查找 B 的 X 坐标,再查找 A 的 X 坐标;FALSE 表示选择精密匹配。计算过程就是 $2438.579 - 2321.652 = 116.927$ 。

(2) 计算 Y 的坐标增量

在单元格 F2 中输入: $=VLOOKUP(RIGHT($D2), A2 : C5, 3, FALSE) - VLOOKUP(LEFT($D2), A2 : C5, 3, FALSE)$ ✓(确认), 单元格 F2 显示: 104.774。

与上列计算相同,只是查找指定的单元格区域的第三列数值,即 Y 坐标值。

若采用拖动,将单元格 E2 中的公式拖动到单元格 F2 中,只需将指定列数修改为 3 即可。公式中采用了半绝对引用,就是为考虑这个拖动而设计的。

(3) 反算坐标方位角

在单元格 G2 中输入: $=MOD(ATAN2(E2, F2), 2 * PI())$ ✓(确认), 单元格 G2 显示: 0.7306359, 这就是 AB 方向以弧度为单位的坐标方位角。

(4) 弧度转换度

在单元格 H2 中输入: $=INT(DEGREES(F2)) + INT(MOD(DEGREES(F2), 1) * 60) / 100 + MOD(DEGREES(F2) * 60, 1) * 3 / 500$ ✓(确认), 单元格 H2 显示: 41.51445。即 AB 的坐标方位角 $\alpha_{AB} = 41^{\circ}51'44''5$ 。

(5) 为了检核弧度转化度分秒的正确性,再将度分秒转化为弧度

在单元格 I2 中输入: $=RADIANS(INT(H2) + INT(MOD(H2, 1) * 100) / 60 + MOD(H2 * 100, 1) / 36)$ ✓(确认), 单元格 I2 显示: 0.7306359。则检核换算正确。

(6) 采用拖动的方法计算其他各方向

选中单元格区域 E2 : I2,用鼠标按住单元格 I2 的拖动柄,拖动至单元格 I7,则在拖动区域 E3 : I7 内,各方向的各项计算一次完成,如图 1.6 所示。这个过程充分体现了 Excel 拖动计算的优越性。

有时为了节省表格中单元格的占用,将以上计算方位角弧度的过程合并成一个公式一次计算出。下面给出这个公式供参考:

在单元格 J2 中输入: $=MOD(ATAN2(VLOOKUP(RIGHT(D2), A2 : C5, 2, FALSE) - VLOOKUP(LEFT(D2), A2 : C5, 2, FALSE), VLOOKUP(RIGHT(D2), A2 : C5, 3, FALSE) - VLOOKUP(LEFT(D2), A2 : C5, 3, FALSE)), 2 * PI())$ ✓(确认), 单元格 J2 显示: 0.7306359。

这就是根据给出的已知数据,直接计算出弧度方位角。当然,这个公式比较长,容易出错,使用时应注意仔细检查。

查找与引用函数在计算过程的各环节中,对数据的流转起到非常重要的作用。